```
y:=sin(3);
```

$\sin(3)$

```
ex:=2*a*sin(x/y);
prog::exprtree(ex)
```

$$2\,a\,\sin\left(\frac{x}{\sin(3)}\right)$$

```
_mult
|
+-- a
|
+-- sin
|   |
|   `-- _mult
|       |
|       +-- x
|       |
|       `-- _power
|           |
|           +-- sin
|           |   |
|           |   `-- 3
|           |
|           `-- -1
|
`-- 2
```

Tree1

```
nops(ex);
```

3

```
op(ex,1);
```

a

```
op(ex,2);
```

$$\sin\left(\frac{x}{\sin(3)}\right)$$

```
op(ex,3);
```

2

```
op(ex,[2,1]);
```

$$\frac{x}{\sin(3)}$$

```
domtype(2);
```

DOM_INT

```
domtype(2/3);
```

DOM_RAT

```
domtype(0.5);
```

DOM_FLOAT

```
domtype(FALSE);
```

DOM_BOOL

```
domtype(ex)
```

DOM_EXPR

```
domtype(x);
```

DOM_IDENT

```
x:=1+I;
```

1 + i

```
domtype(x);
```

DOM_COMPLEX

```
s:="this is a string";
```

"this is a string"

```
domtype(s);
```

DOM_STRING

```
s2:="also this";
```

"also this"

```
s.s2;
```

"this is a stringalso this"

```
s[3..6]
```

"is i"

```
reset();
k:=4;
```

4

```
y.k:=99;
```

99

```
y4;
```

```
99
```

```
f:=-x^2;
```

$$-x^2$$

```
domtype(f)
```

DOM_EXPR

```
g:=x->-x^2;
```

$$x \to -x^2$$

```
domtype(g);
```

DOM_PROC

```
f(2);
```

$$-x(2)^2$$

```
g(2) ;
```

$$-4$$

```
h:=(x,y)->x^2+y^2;
```

$$(x, y) \to x^2 + y^2$$

```
h(1,2);
```

$$5$$

```
myabs:=x->if x>=0 then x else -x end_if;
```

```
x -> (if 0 <= x then x el...
```

```
myabs(3);
```

$$3$$

```
myabs(-3);
```

$$3$$

```
g2:=x->x+2;
```

$$x \to x + 2$$

```
w:=g2@g;
```

$$(x \to x + 2) \circ (x \to -x^2)$$

```
w(2);
```

$$-2$$

```
u:=x-->f;
```

$$x \to -x^2$$

```
u(2);
```

```
-4
```

```
TRUE or UNKNOWN;
```

TRUE

```
TRUE and UNKNOWN;
```

UNKNOWN

```
not bool(1=2);
```

TRUE

```
not bool(1<>2) ;
```

FALSE

```
 set:={a,1,3,x->4};
```

$\{1, 3, a, x \to 4\}$

```
 nops(set);
```

4

```
 set2:={a,3};
```

$\{3, a\}$

```
 set union set2;
```

$\{1, 3, a, x \to 4\}$

```
 set intersect set2;
```

$\{3, a\}$

```
 set minus set2;
```

$\{1, x \to 4\}$

```
 contains(set,1);
```

TRUE

```
 g:=x->-x^2;
```

$x \to -x^2$

```
 map(set2,g);
```

$\{-9, -a^2\}$

```
 g2:=x->bool(x>0);
```

$x \to \text{bool}(0 < x)$

```
a:=-4;
```

-4

```
select(set2,g2);
```

{3}

```
seq:=1,a,TRUE;
```

1, -4, TRUE

```
domtype(seq);
```

DOM_EXPR

```
nops(seq);
```

3

```
op(seq,3);
```

TRUE

```
seq[3];
```

TRUE

```
seq2:=i^2 $ i=1..6;
```

1, 4, 9, 16, 25, 36

```
2 $ 10;
```

2, 2, 2, 2, 2, 2, 2, 2, 2, 2

```
sin(x) $ x in [0,PI,2];
```

0, 0, sin(2)

```
seq3:=seq,seq2;
```

1, -4, TRUE, 1, 4, 9, 16, 25, 36

```
domtype(null());
```

DOM_NULL

```
seq[2]:=2,3;
```

2, 3

```
seq;
```

1, 2, 3, TRUE

```
delete seq[2];
```

```
seq;
```

1, 3, TRUE

```
max(seq2);
```
36

```
min(seq2) ;
```
1

```
lst:=[1,a,TRUE];
```
$[1, -4, \text{TRUE}]$

```
domtype(lst);
```
DOM_LIST

```
nops(lst);
```
3

```
op(lst,3);
```
TRUE

```
lst[3];
```
TRUE

```
lst2:=[];
```
$[]$

```
lst3:=lst.[r,y];
```
$[1, -4, \text{TRUE}, r, y]$

```
[a,b,c]:=[1,2,3];
```
$[1, 2, 3]$

```
a:=1;
```
1

```
b:=2;
```
2

```
[a,b]:=[b,a];
```
$[2, 1]$

```
 a;
```
2

```
b;
```
1

```
contains(lst,1);
```
1

```
map([x,1,0],sin);
```
$[\sin(x), \sin(1), 0]$

```
sort([4,7,-2]);
```

```
[-2, 4, 7]
```

```
select([4,7,-2],g2);
```

```
[4, 7]
```

```
x:=1;
```

```
1
```

```
for i from 1 to 5 do
x:=x+i;
end_for;
```

```
16
```

```
y:=1;
```

```
1
```

```
for i from 1 to 5 step 2 do
y:=y+i;
end_for
```

```
10
```

```
reset();
myFact:= proc(n)
begin ;
f:=1;
for i from 2 to n do
f:=f*i;
end_for;
return(f);
end_proc;
myFact(3)
```

```
proc myFact(n) ... end
```

```
6
```

```
reset();
test1:=proc(n)
begin
 primes:=select([$2..floor(n/2)],isprime);
 lst:=[n $ nops(primes)] - primes;
 good:=select(lst,isprime);
 return (bool(nops(good)>0));
end_proc;
```

```
proc test1(n) ... end
```

```
n:=500;
```

```
500
```

```
l:=[2*i $ i=2..floor(n/2)];
```

```
[4, 6, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34, 36, 38, 40, 42, 44, 46, 48, 50, 52, 54, 56,
  58, 60, 62, 64, 66, 68, 70, 72, 74, 76, 78, 80, 82, 84, 86, 88, 90, 92, 94, 96, 98, 100, 102, 104, 106,
  108, 110, 112, 114, 116, 118, 120, 122, 124, 126, 128, 130, 132, 134, 136, 138, 140, 142, 144, 146,
  148, 150, 152, 154, 156, 158, 160, 162, 164, 166, 168, 170, 172, 174, 176, 178, 180, 182, 184, 186,
  188, 190, 192, 194, 196, 198, 200, 202, 204, 206, 208, 210, 212, 214, 216, 218, 220, 222, 224, 226,
  228, 230, 232, 234, 236, 238, 240, 242, 244, 246, 248, 250, 252, 254, 256, 258, 260, 262, 264, 266,
  268, 270, 272, 274, 276, 278, 280, 282, 284, 286, 288, 290, 292, 294, 296, 298, 300, 302, 304, 306,
  308, 310, 312, 314, 316, 318, 320, 322, 324, 326, 328, 330, 332, 334, 336, 338, 340, 342, 344, 346,
  348, 350, 352, 354, 356, 358, 360, 362, 364, 366, 368, 370, 372, 374, 376, 378, 380, 382, 384, 386,
  388, 390, 392, 394, 396, 398, 400, 402, 404, 406, 408, 410, 412, 414, 416, 418, 420, 422, 424, 426,
  428, 430, 432, 434, 436, 438, 440, 442, 444, 446, 448, 450, 452, 454, 456, 458, 460, 462, 464, 466,
  468, 470, 472, 474, 476, 478, 480, 482, 484, 486, 488, 490, 492, 494, 496, 498, 500]
```

```
select(l,not test1);
```

```
[]
```

```
constructor:=Dom::IntegerMod(7);
```

```
Dom::IntegerMod(7)
```

```
x:=constructor(3);
```

```
3 mod 7
```

```
y:=constructor(5);
```

```
5 mod 7
```

```
domtype(x);
```

```
Dom::IntegerMod(7)
```

```
x+y;
```

```
1 mod 7
```

```
x*y;
```

```
1 mod 7
```

```
x^123;
```

```
6 mod 7
```

```
reset();
p:=poly(1+a*x+3*x^2,[x]);
```

$$\text{poly}(3\,x^2 + a\,x + 1,\ [x])$$

```
domtype(p);
```

```
DOM_POLY
```

```
p|x=2;
```

$$2\,a + 13$$

```
degree(p,x);
```

```
2
```

```
coeff(p,2);
```

$$3$$

```
list:=[[1,0],[a,3],[b,5]];
```

$$[[1, 0], [a, 3], [b, 5]]$$

```
p2:=poly(list,[x]);
```

$$\mathrm{poly}(b\,x^5 + a\,x^3 + 1, [x])$$

```
poly2list(p);
```

$$[[3, 2], [a, 1], [1, 0]]$$

```
p3:=poly(x+1,[x]);
```

$$\mathrm{poly}(x + 1, [x])$$

```
p+p3;
```

$$\mathrm{poly}(3\,x^2 + (a + 1)\,x + 2, [x])$$

```
p*p3;
```

$$\mathrm{poly}(3\,x^3 + (a + 3)\,x^2 + (a + 1)\,x + 1, [x])$$

```
divide(p,p3);
```

$$\mathrm{poly}(3\,x + a - 3, [x]),\ \mathrm{poly}(-a + 4, [x])$$

```
factor(p*p3);
```

$$\mathrm{poly}(x + 1, [x])\ \mathrm{poly}(3\,x^2 + a\,x + 1, [x])$$

```
gcd(p,p3);
```

$$\mathrm{poly}(1, [x])$$

```
D(p);
```

$$\mathrm{poly}(6\,x + a, [x])$$

```
int(p);
```

$$\mathrm{poly}\left(x^3 + \frac{a}{2}\,x^2 + x, [x]\right)$$

```
p4:=poly(4*x+11,[x],Dom::IntegerMod(3));
```

$$\mathrm{poly}(x + 2, [x], \mathrm{Dom::IntegerMod}(3))$$

```
domtype(p4);
```

$$\mathrm{DOM\_POLY}$$

```
p4|x=2;
```

$$1 \bmod 3$$

```
reset();
```

```
A:=matrix([[1,2,3,4],
[a,b,c,d],
[sin(x),cos(x),exp(x),ln(x)]]);
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 \\ a & b & c & d \\ \sin(x) & \cos(x) & e^x & \ln(x) \end{pmatrix}$$

```
Dom::Matrix();
```

Dom::Matrix()

```
v:=matrix([[x1],[x2],[x3],[x4]]);
```

$$\begin{pmatrix} x1 \\ x2 \\ x3 \\ x4 \end{pmatrix}$$

```
A*v;
```

$$\begin{pmatrix} x1 + 2\,x2 + 3\,x3 + 4\,x4 \\ a\,x1 + b\,x2 + c\,x3 + d\,x4 \\ x2\cos(x) + x3\,e^{x} + x4\ln(x) + x1\sin(x) \end{pmatrix}$$

```
A[2,3];
```

c

```
A[2,3]:=4;
```

4

```
A[1..2,2..3];
```

$$\begin{pmatrix} 2 & 3 \\ b & 4 \end{pmatrix}$$

```
transpose(A);
```

$$\begin{pmatrix} 1 & a & \sin(x) \\ 2 & b & \cos(x) \\ 3 & 4 & e^{x} \\ 4 & d & \ln(x) \end{pmatrix}$$

```
diff(A,x);
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ \cos(x) & -\sin(x) & e^{x} & \frac{1}{x} \end{pmatrix}$$

```
int(A,x);
```

$$\begin{pmatrix} x & 2\,x & 3\,x & 4\,x \\ a\,x & b\,x & 4\,x & d\,x \\ -\cos(x) & \sin(x) & e^{x} & x\,(\ln(x) - 1) \end{pmatrix}$$

```
map(A,x->x^2);
```

$$\begin{pmatrix} 1 & 4 & 9 & 16 \\ a^{2} & b^{2} & 16 & d^{2} \\ \sin(x)^{2} & \cos(x)^{2} & e^{2\,x} & \ln(x)^{2} \end{pmatrix}$$

```
constructor:=Dom::Matrix(Dom::Rational);
```

Dom::Matrix(Dom::Rational)

```
A:=constructor(2,3);
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
B:=constructor([[1,2,3],[1,2,3]]);
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 1 & 2 & 3 \end{pmatrix}$$

```
C:=constructor(2,3,(i,j)->i*j);
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 2 & 4 & 6 \end{pmatrix}$$

```
constructor(2,2,[11,12],Diagonal);
```

$$\begin{pmatrix} 11 & 0 \\ 0 & 12 \end{pmatrix}$$

```
constructor::identity(2);
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
constructor2:=Dom::SquareMatrix(2);
```

Dom::SquareMatrix(2)

```
A:=constructor2([[0,y],[x^2,1]]);
```

$$\begin{pmatrix} 0 & y \\ x^2 & 1 \end{pmatrix}$$

```
domtype(A);
```

Dom::SquareMatrix(2)

```
A^(-1);
```

$$\begin{pmatrix} -\frac{1}{x^2 y} & \frac{1}{x^2} \\ \frac{1}{y} & 0 \end{pmatrix}$$

```
exp(A);
```

$$\begin{pmatrix} \frac{\sigma_1-\sigma_2+\sigma_1\sigma_3+\sigma_2\sigma_3}{{\sigma_3}^2} & \frac{\sigma_1-\sigma_2-\sigma_1\left(4yx^2+1\right)+\sigma_2\left(4yx^2+1\right)}{x^2{\sigma_3}^4} \\ -\frac{x^2\sigma_1-x^2\sigma_2}{\sigma_3} & \frac{-\sigma_1+\sigma_2+\sigma_1\sigma_3+\sigma_2\sigma_3}{{\sigma_3}^2} \end{pmatrix}$$

where

$$\sigma_1 = e^{-\frac{\sigma_3}{2}+\frac{1}{2}}$$

$$\sigma_2 = e^{\frac{\sigma_3}{2}+\frac{1}{2}}$$

$$\sigma_3 = \sqrt{4yx^2+1}$$

```
con:=Dom::Matrix(Dom::Rational);
```

Dom::Matrix(Dom::Rational)

```
H:=con(15,15,(i,j)->1/(i+j+1));
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} \\ \frac{1}{4} & \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} \\ \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} \\ \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} \\ \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} \\ \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} \\ \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} \\ \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} \\ \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} \\ \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} & \frac{1}{27} \\ \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} & \frac{1}{27} & \frac{1}{28} \\ \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} & \frac{1}{27} & \frac{1}{28} & \frac{1}{29} \\ \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} & \frac{1}{27} & \frac{1}{28} & \frac{1}{29} & \frac{1}{30} \\ \frac{1}{17} & \frac{1}{18} & \frac{1}{19} & \frac{1}{20} & \frac{1}{21} & \frac{1}{22} & \frac{1}{23} & \frac{1}{24} & \frac{1}{25} & \frac{1}{26} & \frac{1}{27} & \frac{1}{28} & \frac{1}{29} & \frac{1}{30} & \frac{1}{31} \end{pmatrix}$$

```
e:=con(15,1,1);
```

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

```
x:=H^(-1)*e;
```

$$\begin{pmatrix} 2040 \\ -171360 \\ 5290740 \\ -84651840 \\ 814773960 \\ -5121436320 \\ 22086194130 \\ -67310305920 \\ 147241294200 \\ -232016584800 \\ 261018657900 \\ -204434193600 \\ 105867707400 \\ -32574679200 \\ 4508102925 \end{pmatrix}$$

```
DIGITS:=100;
```

100

```
con:=Dom::Matrix(Dom::Rational);
```

Dom::Matrix(Dom::Rational)

```
M:=Dom::Matrix(Dom::Rational);
```

Dom::Matrix(Dom::Rational)

```
H:=con(3,3,(i,j)->1/(i+j+1));
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \\ \frac{1}{4} & \frac{1}{5} & \frac{1}{6} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} \end{pmatrix}$$

```
M::col(H,2);
```

$$\begin{pmatrix} \frac{1}{4} \\ \frac{1}{5} \\ \frac{1}{6} \end{pmatrix}$$

```
M::row(H,2);
```

$$\begin{pmatrix} \frac{1}{4} & \frac{1}{5} & \frac{1}{6} \end{pmatrix}$$

```
M::delCol(H,2);
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{5} \\ \frac{1}{4} & \frac{1}{6} \\ \frac{1}{5} & \frac{1}{7} \end{pmatrix}$$

```
M::delRow(H,2);
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} \end{pmatrix}$$

```
M::matdim(H);
```

[3, 3]

```
M::tr(H);
```

$$\frac{71}{105}$$

```
M::transpose(H);
```

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} \\ \frac{1}{4} & \frac{1}{5} & \frac{1}{6} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} \end{pmatrix}$$

```
M::identity(6);
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

```
A:=M(2,2,[[1,2],[3,1]]);
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 1 \end{pmatrix}$$

```
linalg::det(A);
```

$$-5$$

```
linalg::charpoly(A,y);
```

$$y^2 - 2\,y - 5$$

```
linalg::eigenvalues(A);
```

$$\emptyset$$

```
M:=Dom::Matrix(Dom::Float);
```

Dom::Matrix(Dom::Float)

```
A:=M(2,2,[[1,2],[3,1]]);
```

$$\begin{pmatrix} 1.0 & 2.0 \\ 3.0 & 1.0 \end{pmatrix}$$

```
linalg::eigenvalues(A);
```

{−1.4494897427831780981972840747058913919659474806566701284326925672509603774573150265\
53985943310464023 5, 3.4494897427831780981972840747058913919659474806566701284326925672\
50960377457315026539859433104640235}

```
linalg::eigenvectors(A);
```

[[3.4494897427831780981972840747058913919659474806566701284326925672509603774573150265\
39859433104640235, 1, [[[0.6324555320336758663997787088865437067439110278650433653715\
09705585188877278476442688496216758600 59],
[0.7745966692414833770358530799564799221665843410583181653175147532226966183873958 06\
70385747537173470 36]]]], [−1.4494897427831780981972840747058913919659474806566701284 3\
26925672509603774573150265398594331046402 35, 1, [[[−0.6324555320336758663997787088865\
43706743911027865043365371500970558518887727847644268849621675860059],
[0.7745966692414833770358530799564799221665843410583181653175147532226966183873958 06\
70385747537173470 36]]]]]

```
linalg::nullspace(A);
```

[]

```
a:=matrix([[1,2],[3,5]]);
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 5 \end{pmatrix}$$

```
det(a);
```

$$-1$$

```
numeric::det(a);
```

$$-1.0$$

```
linalg::eigenvalues(a);
```

$$\{3-\sqrt{10},\ \sqrt{10}+3\}$$

```
numeric::eigenvalues(a);
```

[6.16227766016837933199889354443271853371955513932521682685750485279259443863923822134\
4248108379300295, −0.16227766016837933199889354443271853371955513932521682685750485\
2792594438639238221344248108379300295 2]

```
reset();
f:=1/(exp(x^2)+1);
```

$$\frac{1}{e^{x^2}+1}$$

```
g:=diff(f,x);
```

$$-\frac{2\,x\,e^{x^2}}{\left(e^{x^2}+1\right)^2}$$

```
int(g,x);
```

$$\frac{1}{e^{x^2}+1}$$

```
int(g,x=0..PI);
```

$$\frac{1}{e^{\pi^2}+1}-\frac{1}{2}$$

```
p:=exp(-x^2);
```

$$e^{-x^2}$$

```
int(p,x=-0..5);
```

$$\frac{\sqrt{\pi}\,\mathrm{erf}(5)}{2}$$

```
simplify( (exp(x)-1)/(exp(x/2)+1) );
```

$$e^{x/2}-1$$

```
simplify((cos(x))^2+(sin(x))^2);
```

$$1$$

```
limit(sin(x)/x,x=0);
```

$$1$$

```
limit((1+1/n)^n,n=infinity);
```

$$e$$

```
sum(i,i=1..n);
```

$$\frac{n\,(n+1)}{2}$$

```
sum(1/i^2,i=1..infinity);
```

$$\frac{\pi^2}{6}$$

```
product(i^3,i=1..n);
```

$$n!^3$$

```
eqn:={x+y=a,x-a*y=b};
```

$$\{x+y=a,\ x-a\,y=b\}$$

```
solve(eqn,{x,y});
```

$$\begin{cases} \left\{\left[x=\frac{a^2+b}{a+1},\ y=\frac{a-b}{a+1}\right]\right\} & \text{if } a\neq -1 \\ \{[x=-z-1,\ y=z]\} & \text{if } a=-1 \wedge b=-1 \\ \varnothing & \text{if } a=-1 \wedge b\neq -1 \end{cases}$$

```
solve(x*exp(x)=x,{x});
```

$$(x) \in \{(2\,\pi\,k\,\mathrm{i}) \mid k \in \mathbb{Z}\}$$

```
solve(x*exp(x)=x,{x},Real);
```

$$\{[x=0]\}$$

```
die := random(1..6);
```

proc random() ... end

```
die() $ i = 1..20 ;
```

4, 3, 4, 6, 5, 3, 6, 3, 2, 2, 2, 4, 4, 3, 3, 2, 1, 4, 4, 6

```
n := stats::normalRandom(1,3);
```

proc n() ... end

```
n() $ i = 1..20;
```

0.08246332598, 0.01372971472, 0.1060112816, −0.889346086, 4.087416217, 2.103373604, 2.195520634, 1.588855508, 3.039834827, −0.03415436006, −1.401051242, −0.694503791, −0.36683284, 1.362125436, 1.378762055, −0.2803558955, 2.125162477, 2.084038475, 7.247328352, 0.4250666925

```
u := stats::uniformRandom(1,3);
```

proc u() ... end

```
u() $ i = 1..20 ;
```

2.835443688, 2.705667995, 1.040935403, 1.364254937, 2.76328391, 2.516756717, 2.075771544, 2.341294666, 2.536763713, 2.814471874, 1.893579666, 2.702989494, 1.788103892, 1.771497027, 1.548483545, 1.414521647, 2.19508155, 2.455392566, 1.819040605, 2.673157143

```
plot(sin(x));
```




```
plot(sin(x)/x);
```




```
plot(sin(x)/x,x=-1..1);
```




```
plot(x^2+5*y^2,x=-3..3,y=-3..3,#3D);
```




```
plot(x^2-5*y^2,x=-3..3,y=-3..3,#3D);
```




```
plot(-x^2-5*y^2,x=-3..3,y=-3..3,#3D);
```




```
plot(plot::PointList2d([[1,1],[2,2],[3,3]]));
```




```
plot(plot::Polygon2d([[1,1],[2,4],[3,3]]));
```




```
plot(plot::PointList3d([[1,1,1], [1,2,2], [1,3,2], [1,3,4],
[2,1,1], [2,2,3], [2,3.5, 4]],PointSize=5));
```




```
plot(plot::Polygon3d([[1,1,1],[2,4,2],[3,3,1]]));
```




```
plot(plot::Implicit2d(x^3 + x + 2 = y^2,x = -5..5, y = -5..5));
```




```
plot(plot::Implicit3d(x^2+y^2+z^2=1,x=-2..2,y=-2..2,z=-2..2));
```




```
plot(plot::Implicit3d(x^2-y^2+z^2=1,x=-2..2,y=-2..2,z=-2..2));
```




```
plot(plot::Implicit3d(x^2-y^2-z^2=1,x=-2..2,y=-2..2,z=-2..2));
```




```
plot(plot::Implicit3d(-x^2-y^2-z^2=1,x=-2..2,y=-2..2,z=-2..2));
```




```
plot([2*cos(t),sin(t)],t=0..2*PI);
```




```
plot([t*cos(t),t*sin(t)],t=0..2*PI);
```




```
plot([t-sin(t),1-cos(t)],t=0..6*PI) ;
```




```
plot(plot::Polar([1-cos(t),t],t=0..2*PI));
```


y
1.2
1.0
0.8
0.6
0.4
0.2
-2.0
-1.5
-1.0
-0.5
-0.2
x
-0.4
-0.6
-0.8
-1.0
-1.2